

Análisis de la posible evolución de la epidemia de coronavirus COVID-19 por medio de un modelo SEIR

José Manuel Gutiérrez y Juan Luis Varona
Departamento de Matemáticas y Computación
Universidad de La Rioja

15 de marzo de 2020

En matemáticas, modelizar es intentar extraer los aspectos significativamente importantes de una situación real y plasmarlos en forma de expresiones y ecuaciones matemáticas. El objetivo, por supuesto, es poner en marcha la maquinaria matemática para intentar explicar la realidad física o predecirla. Salvo en procesos muy sencillos, lo que se obtenga nunca va a ser exacto, porque el modelo, necesariamente, no puede captar toda la realidad ni todas las variables involucradas, ni incorporar todos los datos iniciales de los que se parte. Pero sí que puede servir para estudiar el proceso y prever, de alguna forma, cómo la introducción de medidas concretas puede hacer que su evolución futura varíe.

En particular, es lo que ocurre con el desarrollo de las epidemias, y es lo que vamos a intentar hacer ahora, mostrando primero el modelo epidemiológico más común (SIR) y otros mejor adaptados a la epidemia de coronavirus COVID-19, como el denominado SEIR, al que incorporaremos variaciones para poder tener en cuenta las medidas de contención que se adoptan. Todos estos modelos requieren el uso de ecuaciones sistemas de ecuaciones diferenciales.

No intentamos hacer una predicción real con números concretos (en particular, para ello necesitaríamos más datos médicos y políticos de los que nosotros disponemos), sino mostrar la influencia de las medidas en el desarrollo de la epidemia. Sólo queremos mostrar las tendencias cualitativas —con una explicación más o menos técnica de las herramientas matemáticas que se usan en los cálculos y las incertidumbres inherentes al proceso, las mismas que se encuentran todos los epidemiólogos—, no dar previsiones cuantitativas de cómo va a ser el futuro de la epidemia.

1. Modelo epidemiológico SIR


Uno de los modelos epidemiológicos más usados es el denominado modelo SIR, que fue propuesto por W. O. Kermack y A. G. McKendrick en 1927.

En una población de tamaño fijo N en la que se ha desatado una epidemia que se propaga mediante contagio, en un tiempo t los individuos pueden estar en tres estados distintos:

- susceptibles $S(t)$,
- infectados $I(t)$,
- recuperados $R(t)$

(lógicamente, $S(t) + I(t) + R(t) = N$).¹

Los susceptibles se pueden infectar cuando entran en contacto con contagiados, con un parámetro β denominado tasa de infección, y cuyo valor depende de si la enfermedad es más o menos contagiosa. Así mismo, los infectados se recuperan con el tiempo, con un parámetro γ denominado tasa de recuperación y que depende del tiempo que suele durar la enfermedad; una vez que se recuperan, los individuos son inmunes, ya no vuelven a ser susceptibles (de hecho, también podemos pensar que no todos los individuos se «recuperan», sino que pueden morir a causa de la enfermedad: ambos tipos de casos están recogidos en $R(t)$, y ya no afectan al desarrollo de la epidemia). Esto se suele representar según el esquema



De acuerdo al comportamiento descrito, la evolución de la epidemia se modela mediante el siguiente sistema de ecuaciones diferenciales:

$$\begin{cases} S'(t) = -\beta S(t)I(t)/N, \\ I'(t) = \beta S(t)I(t)/N - \gamma I(t), \\ R'(t) = \gamma I(t). \end{cases} \quad (1)$$

En el instante inicial, el número de infectados es $I(0) = I_0$; además, $S(0) = N - I_0$ y $R(0) = 0$. A partir de aquí, poniendo en marcha el modelo —es decir, resolviendo el sistema de ecuaciones diferenciales, lo cual se consigue sin dificultades usando los procedimientos numéricos adecuados, aunque no es éste el lugar adecuado para explicar cómo se hace— podemos predecir cuál va a ser el comportamiento de la epidemia. Los parámetros β y γ son, respectivamente, la tasa de transmisión (mide la probabilidad de que un susceptible se infecte cuando entra en contacto con un infectado) y la tasa de recuperación (de tal manera que el periodo medio de recuperación es $1/\gamma$) de la enfermedad, y dependiendo de cuál sea su valor el desarrollo de la epidemia (cuánto dura o el número total de infectados, por ejemplo) puede ser muy distinto.

En una epidemia, un parámetro muy importante es

$$R_0 = \frac{\beta}{\gamma},$$

que se denomina «tasa básica de reproducción», y que representa el número de nuevos infectados producidos por un sólo infectado si toda la población es susceptible (esta idea se ve mejor sin más que escribir $R_0 = \gamma^{-1}/\beta^{-1}$, dado que γ^{-1} y β^{-1} son el periodo medio de recuperación y el tiempo típico entre contactos). Cuanto más pequeño sea R_0 , de manera más lenta evolucionará la epidemia (en la práctica, y para una epidemia real concreta, la observación de la epidemia permite medir R_0 y, a partir de ahí, estimar β).

Veamos un ejemplo de evolución de una enfermedad contagiosa a lo largo de un año (el tiempo lo medimos en meses) sobre una población de $N = 100\,000$ personas, parámetros $\beta = 1$ y $\gamma = 1/4$ (la enfermedad dura, de media, cuatro meses), y un número inicial de infectados $I_0 = 2\,000$. Con estos datos, en la figura 1 podemos observar cómo evolucionaría la epidemia de acuerdo al modelo dado en el sistema (1). (Este ejemplo no intenta representar ninguna enfermedad concreta.)

¹Quizás en español se podían haber elegido nombres más claros, pero, para poder mantener la correspondencia entre el nombre en inglés —idioma en el que se publican prácticamente todos los artículos científicos— y la notación matemática (la inicial del nombre), siempre se usan traducciones literales; lo mismo nos ocurrirá con los «expuestos» $E(t)$ que nos aparecerán más adelante. De hecho, en realidad $R(t)$ suele aludir, en inglés a *retired*, que son los que ya no influyen en la epidemia, bien porque se han recuperado, son inmunes y no contagian, o porque han fallecido; pero no hay una buena traducción que comience por la misma letra, y por eso se usa simplemente «recuperados».




FIGURA 1: Evolución de una epidemia según el modelo SIR durante 12 meses.

Por supuesto, existen muchas variantes de este método. Por ejemplo, podemos pensar que los individuos recuperados ya no son inmunes, sino que vuelven a ser susceptibles, o que los recuperados ya no se pueden contagiar pero pueden contagiar a otros. La casuística —que depende del tipo de enfermedad contagiosa que estemos analizando— es inmensa, y siempre da lugar a variantes sobre el sistema SIR modelizado en (1).


Una forma típica de influir sobre una enfermedad contagiosa es vacunar a la población, o a parte de ella, en un proceso de vacunación que no es inmediato sino que se desarrolla a lo largo del tiempo. Cuando se aplican vacunas al modelo SIR, las nuevas ecuaciones diferenciales que lo modelan son

$$\begin{cases} S'(t) = -\beta S(t)I(t)/N - V(t), \\ I'(t) = \beta S(t)I(t)/N - \gamma I(t), \\ R'(t) = \gamma I(t), \end{cases}$$

donde la función $V(t)$ surge del proceso de vacunación concreto que se haya aplicado. Desde luego, la disponibilidad de una vacuna puede cambiar completamente el proceso de evolución de una epidemia pero, lamentablemente, no es el caso del COVID-19 que nos ocupa ahora.

2. Modelo epidemiológico SEIR

El modelo SIR no se adapta bien al comportamiento de la epidemia del coronavirus dado que, en esta enfermedad, conviene también tener en cuenta a una nueva clase de individuos que portan la enfermedad pero que, al hallarse en su periodo de incubación, no muestran síntomas y aún no pueden infectar a otros. A dichos individuos se los denomina «expuestos», y el número de ellos en el tiempo t se denota mediante $E(t)$ (hay que ser precisos con lo que estamos denotando: si un individuo no presenta síntomas pero sí puede contagiar a otros lo contabilizamos en $I(t)$, no en $E(t)$). Este modelo se denomina SEIR y, en él, hay un nuevo parámetro σ , de tal forma que $1/\sigma$ es el tiempo promedio de incubación. En el modelo SEIR, el flujo entre los distintos grupos (que de nuevo deben cumplir $S(t) + E(t) + I(t) + R(t) = N$) se puede representar mediante



Las ecuaciones diferenciales del modelo SEIR básico son

$$\begin{cases} S'(t) = -\beta S(t)I(t)/N, \\ E'(t) = \beta S(t)I(t)/N - \sigma E(t), \\ I'(t) = \sigma E(t) - \gamma I(t), \\ R'(t) = \gamma I(t), \end{cases} \quad (2)$$

y la tasa básica de reproducción sigue siendo $R_0 = \beta/\gamma$.

3. Modelo epidemiológico para el COVID-19 con medidas de contención

En el caso del coronavirus, y en aras de describir mejor el comportamiento de la enfermedad y estudiar su propagación, diversos estudios han tenido también en cuenta otros tipos de individuos, como los que siempre permanecen asintomáticos (que sí pueden afectar a otros) o los que nunca se infectan (es lo que parece ocurrir con los niños), y más parámetros relacionados con las cuarentenas y los periodos de diagnóstico. Por supuesto, esto hace que los sistemas tengan más ecuaciones y más parámetros; pero, si conociéramos con exactitud el valor de todos los parámetros, el análisis no sería mucho más difícil. Como el valor de los parámetros no suele ser fácil de estimar, introducir muchos y complicar las ecuaciones no suele tener mucha utilidad práctica, así que seguiremos con el método SEIR dado en (2).² Además, bastantes de los estudios ya publicados introducen funciones o ecuaciones adicionales para tener en cuenta el comportamiento del inicio de la epidemia en Wuhan en diciembre de 2019, cuando los humanos se infectaban directamente del contacto con animales, así como los desplazamientos extraordinarios de población fruto de las festividades chinas de la época. Prescindiremos aquí de todo eso, y simplificaremos hasta quedarnos con lo que nos parece razonable.

Como acabamos de comentar, lo más importante para poder estudiar el comportamiento de una epidemia es conocer el valor de los parámetros en ella involucrados: β , γ y σ en el caso del modelo SEIR.

Para el coronavirus COVID-19, y siguiendo lo publicado en [1], $\gamma^{-1} = 5$ días y $\sigma^{-1} = 7$ días (estos números son medias aproximadas, y pueden variar bastante de persona a persona). Lo más complicado es estimar β ; en particular, porque no se sabe cuántas personas infectadas asintomáticas hay que puedan estar infectando a otros (conocer esto requeriría haber hecho pruebas diarias a una parte importante de la población, pero, claro, los recursos y el tiempo son limitados, y sólo se analiza si están enfermos o no de coronavirus los que presentan síntomas).

Diversos estudios sobre el comportamiento de la enfermedad en China antes de que se adoptaran medidas drásticas de aislamiento de la población (como el ya citado [1]) sugieren que el valor de β podría estar entre 0.59 y 1.68 (en unidades días⁻¹), lo cual daría un R_0 entre 2.95 y 8.4, en ambos casos bastante elevado (un reciente artículo en la revista *Lancet* [3] toma $R_0 = 2.68$).

Con estos parámetros $\gamma = 1/5$, $\sigma = 1/7$ y $\beta = 1$, y suponiendo que tenemos una población de 100 000 personas, en la que hay un infectado al principio, la evolución de la epidemia modelada tal como se indica en (2) durante 120 días sería tal como se muestra en la figura 2. Con

²Una modificación que quizás útil podía ser cambiar $S(t)I(t)/N$ por $S(t)E(t)/N$ en las dos primeras ecuaciones de (2), de manera que sean los asintomáticos los que contagian, pues los sintomáticos son muy responsables y están muy bien aislados y ya no tienen contacto con nadie (también se podía incluir ambos tipos sumandos con distintas tasas de transmisión); pero, claro, eso requeriría una reestimación de los parámetros γ y σ que controlan cuánto tiempo permanece un individuo en cada situación.

otros parámetros no muy distintos los valores concretos de las funciones $S(t)$, $E(t)$, $I(t)$ y $R(t)$ variarían, pero el aspecto de las gráficas sería similar.




FIGURA 2: Evolución de una epidemia de COVID-19 según el modelo SEIR durante 120 días.

Hasta ahora hemos considerado que la tasa de infección β es constante. Pero este parámetro se puede ajustar artificialmente si se adoptan medidas de contención (protección y aislamiento), y si la población las acepta y las cumple. Así, se han llevado a cabo diversos estudios con parámetros β que van variando en el tiempo. Por ejemplo (véase [2]), con una función decreciente respecto al tiempo de la forma

$$\beta(t) = \beta_0 \left((c_0 - c_b)e^{-r_1 t} + c_b \right)$$

donde β_0 sería la tasa de infección sin medidas de contención y las constantes c_0 y c_b aluden a las tasas de contacto (fruto de los aislamientos de la población); o, también (véase [1]),

$$\beta(t) = \beta_0 (1 - \alpha(t)) \left(1 - \frac{D(t)}{N} \right)^\kappa, \quad (3)$$

donde β_0 es la tasa de infección sin medidas, $\alpha(t)$ (con valores en el intervalo $[0, 1]$) es el resultado de las acciones gubernamentales, $D(t)$ es la sensación pública de riesgo como consecuencia de los casos críticos y muertes conocidos, y κ mide la intensidad de la reacción de los individuos. Típicamente, $\alpha(t)$ es una función constante a trozos (las medidas se toman en momentos concretos), y algunas estimaciones (citadas en [1]) sugieren que se puede tomar $\kappa = 1117.3$ (aunque parece un tanto ridículo tomar un valor con cinco dígitos de precisión, ese es el valor obtenido en estudios previos para otras situaciones y, a falta de otra información, es lo que ellos usan). La idea del factor $(1 - D(t)/N)^\kappa$ con un κ elevado es que, cuando la preocupación es mucha, el factor es muy cercano a 0, la gente se aísla incluso voluntariamente y $\beta(t)$ es muy pequeño; por el contrario, el factor es cercano a 1 y tiene escasa influencia si la preocupación social es poca. Como la realidad político-social china y la occidental no es la misma, difícilmente se puede pensar que podemos usar sus mismos parámetros, pero sí sus mismas ideas de modelización. Como estimación para $D(t)$ nosotros podemos pensar que el 5% de los casos son graves y, así (sin necesidad de complicar excesivamente las ecuaciones), tomar $D(t) = 0.05I(t)$ y un κ elevado pero más moderado, digamos, $\kappa = 100$.

Si desde el principio se hubiesen tomado medidas de contención, la evolución de la epidemia hubiera sido muy distinta. Por ejemplo, si, en (3) con $\beta_0 = 1$, $\alpha(t)$ hubiera tenido el valor constante 0.5, la evolución de la epidemia hubiera sido como se muestra en la figura 3.




FIGURA 3: Evolución de una epidemia de COVID-19 según el modelo SEIR con medidas de contención aplicadas desde el principio, durante 240 días.

Reiteramos que todo esto son estimaciones, es imposible conocer el futuro, pero sí se puede analizar cómo las medidas afectan a la evolución de la epidemia; en particular, los cambios en el valor de α . Con esto vamos a mostrar tres escenarios, mostrando cómo evolucionaría la epidemia a partir de varios α . En concreto, en la figura 4 se compara cómo cambiaría la evolución de $S(t)$, $E(t)$, $I(t)$ y $R(t)$ si no se aplican medidas de contención (es decir, como en la figura 2) o si estas medidas se comenzaran a aplicar el día 20, es decir con

$$\beta(t) = \begin{cases} \beta_0, & \text{si } t < 20, \\ \beta_0(1 - \alpha)(1 - 0.05I(t)/N)^\kappa, & \text{si } t \geq 20, \end{cases}$$

con $\beta_0 = 1.0$ y $\kappa = 100$. A su vez, analizamos dos valores de α distintos: $\alpha = 0.4$ (medidas pequeñas) y $\alpha = 0.7$ (medidas serias). Está claro que las medidas de contención son realmente efectivas, que permiten suavizar el «pico» y que lo alejan en el tiempo. También, que el efecto de las medidas no es inmediato, sino que tarda días en ponerse de manifiesto. Qué α va a ser el que se obtiene con unas medidas concretas tomadas por un gobierno es imposible de conocer de antemano, sólo podría ser estimado a posteriori a la vista de lo que haya ocurrido; pero, sea cual sea, su efecto va a ser positivo, y, si en la evaluación diaria de los datos epidemiológicos se ve que no es suficiente para lo que se pretende conseguir, se pueden endurecer las medidas.

Los programas que han permitido construir las gráficas de evolución de la epidemia de COVID-19 están hechos con Mathematica, y se muestran en el Anexo.

Nota. Para concluir, citemos que en [4] plantean un modelo basado no en ecuaciones diferenciales sino en ecuaciones en diferencias, y tienen en cuenta que, en el COVID-19, los asintomáticos pueden contagiar a otros. En principio, las ecuaciones en diferencias modelan peor las situaciones reales (de hecho, suelen ser una aproximación de las ecuaciones diferenciales), así que sería mejor tenerlo en forma de ecuaciones diferenciales. Esto se puede hacer sin dificultades, pero los experimentos que hemos realizado parece dar un comportamiento poco realista para un número elevado de días, así que hemos descartado su uso para lo que aquí hemos expuesto.




FIGURA 4: Comparativa de las evolución de $S(t)$, $E(t)$, $I(t)$ y $R(t)$ en una epidemia de COVID-19 durante 240 días, según se apliquen o no medidas de contención.

Referencias

- [1] QIANYING LIN, SHI ZHAO, DAOZHOU GAO, YIJUN LOU, SHU YANG, SALIHU S. MUSA, MAGGIE H. WANG, YONGLI CAI, WEIMING WANG, LIN YANG, DAIHAI HE, A conceptual model for the coronavirus disease 2019 (COVID-19) outbreak in Wuhan, China with individual reaction and governmental action, *International Journal of Infectious Diseases* **93** (2020), 211–216. Publicado: March 04, 2020. [https://www.ijidonline.com/article/S1201-9712\(20\)30117-X/fulltext](https://www.ijidonline.com/article/S1201-9712(20)30117-X/fulltext)
- [2] BIAO TANG, NICOLA LUIGI BRAGAZZI, QIAN LI, SANYI TANG, YANNI XIAO, JIANHONG WU, An updated estimation of the risk of transmission of the novel coronavirus (2019-nCov), *Infectious Disease Modelling* **5** (2020), 248–255. Publicado: February 11, 2020. <https://www.sciencedirect.com/science/article/pii/S246804272030004X>
- [3] JOSEPH T. WU, KATHY LEUNG, GABRIEL M. LEUNG, Nowcasting and forecasting the potential domestic and international spread of the 2019-nCoV outbreak originating in Wuhan, China: a modelling study, *Lancet* **395** (2020), no. 10225, 689–697. Publicado: Jan 31, 2020. [https://www.thelancet.com/journals/lancet/article/PIIS0140-6736\(20\)30260-9/fulltext](https://www.thelancet.com/journals/lancet/article/PIIS0140-6736(20)30260-9/fulltext)
- [4] ZHOU TANG, XIANBIN LI, HOUQIANG LI, Prediction of New Coronavirus Infection Based on a Modified SEIR Model, *MedRxiv* (The preprint server for health sciences) 2020.03.03.20030858. Publicado: March 06, 2020. <https://www.medrxiv.org/content/10.1101/2020.03.03.20030858v1>

Anexo

Modelo SEIR para el COVID-19 con y sin medidas de contención

José Manuel Gutiérrez y Juan Luis Varona

Departamento de Matemáticas y Computación,
Universidad de La Rioja

In[1]:= **\$Version**

Out[1]:= 12.0.0 for Mac OS X x86 (64-bit) (April 7, 2019)

Ejemplo de evolución de la epidemia sin medidas

In[2]:= **Clear[nN, beta0, gamma, sigma, iI0, sol, sS, eE, iI, rR]**

In[3]:= **nN = 100 000; (* Estudio para 100 mil personas *)**

beta0 = 1.0;

gamma = 1 / 5; (* la enfermedad dura, de media, cinco días *)


sigma = 1 / 7; (* el periodo asintomático dura, de media, siete días *)

iI0 = 1; (* un infectado al principio *)

(* tiempo t en días *)

In[8]:= **sol = NDSolve[{sS'[t] == -beta0 * sS[t] * iI[t] / nN,
eE'[t] == beta0 * sS[t] * iI[t] / nN - sigma * eE[t],
iI'[t] == sigma * eE[t] - gamma * iI[t],
rR'[t] == gamma * iI[t],
sS[0] == nN - iI0, eE[0] == iI0, iI[0] == 0, rR[0] == 0},
{sS, eE, iI, rR}, {t, 0, 240}];**

In[9]:= **Plot[Evaluate[{sS[t], eE[t], iI[t], rR[t]} /. sol], {t, 0, 120},
PlotRange -> All, AxesOrigin -> {0, 0}, AxesLabel -> {"días"},
PlotLegends -> {"susceptibles", "expuestos", "infectados", "recuperados"}]**



Ejemplo con medidas de contención y aislamiento desde el principio (ya no se puede hacer, claro)

```


In[10]:= Clear[nN, beta0, gamma, sigma, betam, alpha, kappa, iI0, solm, sSm, eEm, iIm, rRm]
In[11]:= nN = 100 000; (* Estudio para 100 mil personas *)
        beta0 = 1.0;
        kappa = 1117.3; (* Lo que pone en el artículo de los chinos,
        pero me parece mucho *)
        kappa = 100;
        gamma = 1/5; (* la enfermedad dura, de media, cinco días *)
        sigma = 1/7; (* el periodo asintomático dura, de media, siete días *)
        iI0 = 1; (* un infectado al principio *)
        (* tiempo t en días *)

In[18]:= alpha[t_] := 0.5; (* Constante desde el principio *)
        betam[t_] := beta0 * (1 - 0.5) * (1 - 0.05 iIm[t] / nN) ^ kappa

In[20]:= solm = NDSolve[{sSm'[t] == -betam[t] * sSm[t] * iIm[t] / nN,
        eEm'[t] == betam[t] * sSm[t] * iIm[t] / nN - sigma * eEm[t],
        iIm'[t] == sigma * eEm[t] - gamma * iIm[t],
        rRm'[t] == gamma * iIm[t],
        sSm[0] == nN - iI0, eEm[0] == iI0, iIm[0] == 0, rRm[0] == 0},
        {sSm, eEm, iIm, rRm}, {t, 0, 240}];

In[21]:= Plot[Evaluate[{sSm[t], eEm[t], iIm[t], rRm[t]} /. solm], {t, 0, 240},
        PlotRange -> All, AxesOrigin -> {0, 0}, AxesLabel -> {"días",},
        PlotLegends -> {"susceptibles", "expuestos", "infectados", "recuperados"}]

```



Comparativa de la evolución al introducir medidas de contención un día concreto (tiempo en días)

El primer contagiado en España fue el 1 de febrero (un alemán en la Gomera que fue aislado), y el segundo un británico en Palma de Mallorca el 9 de febrero (también aislado). El 25 de febrero se

detectan los primeros casos en la península, en Madrid, Barcelona y Castellón (posiblemente, el virus estaba antes pero sin detectar).

Las universidades madrileñas y de La Rioja se cierran el 11 de marzo. El estado de alarma se decreta el 14 de marzo, y la prohibición de salir de casa sin motivo justificado entra en vigor el 16 de marzo.

```

In[22]:= Clear[nN, beta0, gamma, sigma, betam04, betam07,
  alpha, kappa, iI0, sol, solm04, solm07, sSm, eEm, iIm, rRm]

In[23]:= nN = 100 000; (* Estudio para 100 mil personas *)
beta0 = 1.0;
kappa = 1117.3; (* Lo que pone en el artículo de los chinos,
pero me parece mucho *)
kappa = 10;
kappa = 100;
gamma = 1 / 5; (* la enfermedad dura, de media, cinco días *)
sigma = 1 / 7; (* el periodo asintomático dura, de media, siete días *)
iI0 = 1; (* un infectado al principio *)
(* tiempo t en días *)

In[31]:= t0 = 20; (* El día 20 se introducen las medidas *)
betam04[t_] := If[t < t0, beta0, beta0 * (1 - 0.4) * (1 - 0.05 iIm[t] / nN) ^ kappa];
betam07[t_] := If[t < t0, beta0, beta0 * (1 - 0.7) * (1 - 0.05 iIm[t] / nN) ^ kappa];

In[34]:= sol = NDSolve[{sS'[t] == -beta0 * sS[t] * iI[t] / nN,
  eE'[t] == beta0 * sS[t] * iI[t] / nN - sigma * eE[t],
  iI'[t] == sigma * eE[t] - gamma * iI[t],
  rR'[t] == gamma * iI[t],
  sS[0] == nN - iI0, eE[0] == iI0, iI[0] == 0, rR[0] == 0},
  {sS, eE, iI, rR}, {t, 0, 240}];

In[35]:= solm04 = NDSolve[{sSm'[t] == -betam04[t] * sSm[t] * iIm[t] / nN,
  eEm'[t] == betam04[t] * sSm[t] * iIm[t] / nN - sigma * eEm[t],
  iIm'[t] == sigma * eEm[t] - gamma * iIm[t],
  rRm'[t] == gamma * iIm[t],
  sSm[0] == nN - iI0, eEm[0] == iI0, iIm[0] == 0, rRm[0] == 0},
  {sSm, eEm, iIm, rRm}, {t, 0, 240}];


In[36]:= solm07 = NDSolve[{sSm'[t] == -betam07[t] * sSm[t] * iIm[t] / nN,
  eEm'[t] == betam07[t] * sSm[t] * iIm[t] / nN - sigma * eEm[t],
  iIm'[t] == sigma * eEm[t] - gamma * iIm[t],
  rRm'[t] == gamma * iIm[t],
  sSm[0] == nN - iI0, eEm[0] == iI0, iIm[0] == 0, rRm[0] == 0},
  {sSm, eEm, iIm, rRm}, {t, 0, 240}];

```

```

In[37]:= Plot[Evaluate[{{sS[t] /. sol, sSm[t] /. solm04, sSm[t] /. solm07}},
  {t, 0, 240}, PlotRange -> All, AxesOrigin -> {0, 0}, AxesLabel -> {"días"},
  Epilog -> {Text["día " <> ToString[t0], {t0, 0}, {0, -2}], PointSize[0.03],
  Red, Point[{t0, 0}]}], PlotLegends -> {"susceptibles sin medidas",
  "susceptibles con medidas pequeñas", "susceptibles con medidas serias"}]


```




```

In[38]:= Plot[Evaluate[{{eE[t] /. sol, eEm[t] /. solm04, eEm[t] /. solm07}},
  {t, 0, 240}, PlotRange -> All, AxesOrigin -> {0, 0}, AxesLabel -> {"días"},
  Epilog -> {Text["día " <> ToString[t0], {t0, 0}, {0, -2}],
  PointSize[0.03], Red, Point[{t0, 0}]}], PlotLegends -> {"expuestos sin medidas",
  "expuestos con medidas pequeñas", "expuestos con medidas serias"}]


```



```
In[39]:= Plot[Evaluate[{{iI[t] /. sol, iIm[t] /. solm04, iIm[t] /. solm07}},
  {t, 0, 240}, PlotRange -> All, AxesOrigin -> {0, 0}, AxesLabel -> {"días",},
  Epilog -> {Text["día " <> ToString[t0], {t0, 0}, {0, -2}],
  PointSize[0.03], Red, Point[{t0, 0}]}},
  PlotLegends -> {"infectados sin medidas", "infectados con medidas pequeñas",
  "infectados con medidas serias"}]
```



```
In[40]:= Plot[Evaluate[{{rR[t] /. sol, rRm[t] /. solm04, rRm[t] /. solm07}},
  {t, 0, 240}, PlotRange -> All, AxesOrigin -> {0, 0}, AxesLabel -> {"días",},
  Epilog -> {Text["día " <> ToString[t0], {t0, 0}, {0, -2}], PointSize[0.03],
  Red, Point[{t0, 0}]}}, PlotLegends -> {"recuperados sin medidas",
  "recuperados con medidas pequeñas", "recuperados con medidas serias"}]
```



Modelización alternativa usando ecuaciones con retardo (sin medidas de contención)

```
In[41]:= Clear[nN, beta0, gamma, sigma, iI0, solretardo, sS, eE, iI, rR]
```

Modelo adaptado del artículo

«Prediction of New Coronavirus Infection Based on a Modified SEIR Model»,

de Zhou Tang, Xianbin Li, Houqiang Li,

medRxiv preprint, doi: <https://doi.org/10.1101/2020.03.03.20030858>.

Lo que sale con este modelo no tiene mucho sentido pues, como se ve en los dibujos, salen

números de personas mayores del total $nN=100000$, o negativos. Así que no parece útil para periodos largos de tiempo (varios meses), y prescindimos de él.
(Quizás con otros parámetros hubiese funcionado mejor)

```
In[42]:= nN = 100 000; (* Estudio para 100 mil personas *)
beta0 = 1.0;
gamma = 1 / 5; (* la enfermedad dura, de media, cinco días *)
sigma = 1 / 7; (* el periodo asintomático dura, de media, siete días *)
iI0 = 1; (* un infectado al principio *)
(* tiempo t en días *)

In[47]:= solretardo = NDSolve[{sS'[t] == -beta0 * sS[t] * eE[t] / nN,
  eE'[t] == beta0 * sS[t] * eE[t] / nN - sigma * eE[t - 7],
  iI'[t] == sigma * eE[t - 7] - gamma * iI[t],
  rR'[t] == gamma * iI[t],
  sS[t / ; t <= 0] == nN - iI0, eE[t / ; t <= 0] == iI0,
  iI[t / ; t <= 0] == 0, rR[t / ; t <= 0] == 0}, {sS, eE, iI, rR}, {t, 0, 90}];

In[48]:= Plot[Evaluate[{sS[t], eE[t], iI[t], rR[t]} /. solretardo],
  {t, 0, 90}, PlotRange -> All, AxesOrigin -> {0, 0}, AxesLabel -> {"días"},
  PlotLegends -> {"susceptibles", "expuestos", "infectados", "recuperados"}]
```

